
Option Explicit

Dim awert As Integer, bwert As Integer, cwert As Integer

Dim Ueberschrift, Bezeichnung, Bezeichnungwert As String

Dim Kunde(100) As String

Dim wert(100) As Double

Dim AnzahlSaetze As Integer

Dim Geaendert As Boolean

Dim Summe As Double

Dim Gespeichert As Boolean

Private Type Satz

    Kunde As String

    Umsatz As Double

    Prozent As Double

End Type

Dim Sortiert(100) As Satz

Private Sub menu_an_aus()

 mnuDatendrucken.Enabled = Not (AnzahlSaetze = 0)

 mnuDateiLöschen.Enabled = Not (AnzahlSaetze = 0)

 mnuDatenspeichern.Enabled = Not (AnzahlSaetze = 0)

End Sub

Private Sub Form_Initialize()

    lblA.Caption = "A"

    txtA.Text = "70"

    awert = 70

    lblB.Caption = "B"

    txtB.Text = "20"

    bwert = 20

    lblC.Caption = "C"

    txtC.Text = "10"

    cwert = 10

    txtC.Enabled = False

    lblGewichtung.Caption = "Gewichtung"

    lblKundenanalyse.Caption = "Analyse"

    lblKundenname.Visible = False

    lblUmsatz.Visible = False

    txtKundenname.Visible = False

    txtUmsatz.Visible = False

    cmduebernahme.Caption = "Übernehmen"

    cmduebernahme.Visible = False

    cmdabbrechen.Caption = "Abbrechen"

    cmdabbrechen.Visible = False

    Geaendert = False

    grdGrid.Cols = 4

    Gespeichert = False

    lblKundenname.Caption = "Kundenname"

    lblUmsatz.Caption = "Umsatz"

    menu_an_aus

End Sub

Private Sub Form_Unload(Cancel As Integer)

Dim Antwort As Integer

If Geaendert = True Then

Antwort = MsgBox("Sie haben Ihre Daten geändert!" & Chr(13) & "Wollen Sie speichern?" _

       , vbYesNoCancel, "Beenden")

If Antwort = vbYes Then

    mnuDatenspeichern_Click

    If Gespeichert Then

        Cancel = 0

    Else

        Cancel = 1

    End If

    ElseIf Antwort = vbNo Then

        Cancel = 0

        Else

        Cancel = 1

    End If

End If

End Sub

Private Sub mnuBeenden_Click()

    Unload Me

End Sub

Private Sub mnuDateiLöschen_Click()

Dim Antwort As Integer

Dim AnzReihen As Integer

Dim I As Integer

Antwort = vbYes

If Geaendert = True Then Antwort = MsgBox("Sie haben Ihre Daten geändert!" & Chr(13) & "Wollen Sie wirklich löschen?", vbYesNo, "Daten löschen")

If Antwort = vbYes Then

    AnzahlSaetze = 0

    menu_an_aus

    grdGrid.FixedRows = 0

    AnzReihen = grdGrid.Rows - 1

    For I = AnzReihen To 1 Step -1

        grdGrid.RemoveItem (I)

    Next I

    Bezeichnung = ""

    Bezeichnungwert = ""

    grdGrid.TextMatrix(0, 0) = Bezeichnung

    grdGrid.TextMatrix(0, 1) = Bezeichnungwert

End If

End Sub

Private Sub mnuDatendrucken_Click()

    Dim Druckzeile As String

    Dim Druckspalte As String

    Dim Zeile As Integer

    Dim Spalte As Integer

    dlgdialog.Flags = cdlPDReturnDC + cdlPDNoPageNums

    dlgdialog.ShowPrinter

    Printer.Print ""

    For Zeile = 0 To AnzahlSaetze - 1

        Druckzeile = "  "

        Druckspalte = grdGrid.TextMatrix(Zeile, 0)

          Do While Len(Druckspalte) <= 15

            Druckspalte = Druckspalte & " "

            Loop

        Druckzeile = Druckzeile & Druckspalte

        For Spalte = 1 To 3

            Druckspalte = grdGrid.TextMatrix(Zeile, Spalte)

            Do While Len(Druckspalte) <= 13

            Druckspalte = " " & Druckspalte

            Loop

            Druckzeile = Druckzeile & Druckspalte

        Next Spalte

    Druckzeile = Druckzeile & vbCrLf

    Printer.Print Druckzeile

Next Zeile

Printer.EndDoc

End Sub

Private Sub mnuDateneingeben_Click()

    txtKundenname.Text = ""

    txtUmsatz.Text = ""

    lblKundenname.Visible = True

    lblUmsatz.Visible = True

    txtKundenname.Visible = True

    txtUmsatz.Visible = True

    cmduebernahme.Visible = True

    cmdabbrechen.Visible = True

    txtKundenname.SetFocus

    mnuDatei.Enabled = False

End Sub

Private Sub cmdabbrechen_Click()

    mnuDatei.Enabled = True

    lblKundenname.Visible = False

    lblUmsatz.Visible = False

    txtKundenname.Visible = False

    txtUmsatz.Visible = False

    cmduebernahme.Visible = False

    cmdabbrechen.Visible = False

End Sub

Private Sub cmduebernahme_Click()

Dim fehler As Boolean

fehler = False

If AnzahlSaetze < 100 Then

   If txtUmsatz.Text = "" Then

     MsgBox ("Bitte Umsatz eingeben!")

     fehler = True

     txtUmsatz.SetFocus

    ElseIf Not IsNumeric(txtUmsatz.Text) Then

     MsgBox ("Bitte richtigen Umsatz eingeben!")

     fehler = True

     txtUmsatz.SetFocus

    ElseIf Val(txtUmsatz.Text) <= 0 Then

     MsgBox ("Bitte einen zulässigen Umsatz eingeben!")

     fehler = True

     txtUmsatz.SetFocus

   End If

   txtKundenname.Text = Trim(txtKundenname.Text)

   If txtKundenname.Text = "" And Not fehler Then

     MsgBox ("Bitte Kundenname eingeben!")

     fehler = True

     txtKundenname.SetFocus

   End If

Else

    MsgBox ("Ich bin voll. Danke mir reicht es!")

    fehler = True

End If

If fehler = False Then

  Sortiert(AnzahlSaetze).Kunde = txtKundenname.Text

  Sortiert(AnzahlSaetze).Umsatz = txtUmsatz.Text

  AnzahlSaetze = AnzahlSaetze + 1

  menu_an_aus

  Sortieren

  cmdabbrechen_Click

   Geaendert = True

End If

End Sub

Private Sub mnuDateneinlesen_Click()

    Dim dateiname As String

    Dim puffer As String

    Dim Zaehler As Integer

    dlgdialog.Flags = cdlOFNOverwritePrompt

    dlgdialog.Filter = "abc| *.abc"

    dlgdialog.filename = ""

    dlgdialog.ShowOpen

    dateiname = dlgdialog.filename

    If dateiname <> "" Then

        Open dateiname For Input As #1

        Input #1, Ueberschrift

        Input #1, Bezeichnung

        Input #1, Bezeichnungwert

        Zaehler = 0

       While Not EOF(1)

            Input #1, Sortiert(Zaehler).Kunde

            Input #1, Sortiert(Zaehler).Umsatz

            Zaehler = Zaehler + 1

        Wend

        Close #1

        AnzahlSaetze = Zaehler

        Sortieren

        menu_an_aus

      Else

        MsgBox "Sie haben keine gültige Datei gewählt"

    End If

End Sub

Private Sub mnuDatenspeichern_Click()

    Dim dateiname As String

    Dim puffer As String

    Dim Zaehler As Integer

If AnzahlSaetze > 0 Then

    dlgdialog.Flags = cdlOFNOverwritePrompt

    dlgdialog.Filter = "abc| *.abc"

    dlgdialog.filename = ""

    dlgdialog.ShowSave

    dateiname = dlgdialog.filename

    If dateiname <> "" Then

        Open dateiname For Output As #1

        Write #1, Ueberschrift

        Write #1, Bezeichnung

        Write #1, Bezeichnungwert

        For Zaehler = 0 To AnzahlSaetze - 1

            Write #1, Sortiert(Zaehler).Kunde

            Write #1, Sortiert(Zaehler).Umsatz

        Next Zaehler

        Close #1

        Gespeichert = True

        Geaendert = False

      Else

        MsgBox "Sie haben keine gültige Datei gewählt"

        Gespeichert = False

    End If

End If

End Sub

Private Sub mnuHelp_Click()

    MsgBox "Programm zur ABC-Analyse von Daten!"

End Sub

Sub Sortieren()

Dim I, j As Integer

Dim vergleich As Satz

Summe = 0

For I = 0 To AnzahlSaetze - 1

    Summe = Summe + Sortiert(I).Umsatz

Next I

    For I = 0 To AnzahlSaetze - 1

        For j = I + 1 To AnzahlSaetze - 1

            If Sortiert(j).Umsatz > Sortiert(I).Umsatz Then

                vergleich = Sortiert(j)

                Sortiert(j) = Sortiert(I)

                Sortiert(I) = vergleich

            End If

        Next j

    Next I

For I = 0 To AnzahlSaetze - 1

    Sortiert(I).Prozent = Sortiert(I).Umsatz / Summe * 100

Next I

Grid_Fuellen

End Sub

Sub Grid_Fuellen()

Dim I As Integer

Dim GesProz As Double

Dim AnzReihen As Integer

grdGrid.FixedRows = 0

AnzReihen = grdGrid.Rows - 1

For I = AnzReihen To 1 Step -1

    grdGrid.RemoveItem (I)

Next I

GesProz = 0

If Bezeichnung = "" Then _

   Bezeichnung = InputBox("Geben Sie eine Überschrift für die erste Spalte ein!", _

   "Überschrift erstes Feld")

If Bezeichnungwert = "" Then _

   Bezeichnungwert = InputBox("Geben Sie eine Überschrift für die zweite Spalte ein!", _

   "Überschrift zweites Feld")

grdGrid.TextMatrix(0, 0) = Bezeichnung

grdGrid.TextMatrix(0, 1) = Bezeichnungwert

grdGrid.TextMatrix(0, 2) = "%-Satz"

grdGrid.TextMatrix(0, 3) = "%-Satz kum."

For I = 1 To AnzahlSaetze

    grdGrid.AddItem ""

    GesProz = GesProz + Sortiert(I - 1).Prozent

    If I = 1 Then

       grid_color I, vbGreen

     Else

       Select Case GesProz

         Case Is <= awert

              grid_color I, vbGreen

         Case Is <= bwert + awert

              grid_color I, vbYellow

         Case Else

              grid_color I, vbRed

        End Select

    End If

    grdGrid.TextMatrix(I, 0) = Sortiert(I - 1).Kunde

    grdGrid.TextMatrix(I, 1) = Sortiert(I - 1).Umsatz

    grdGrid.TextMatrix(I, 2) = Format(Sortiert(I - 1).Prozent, "0.00")

    grdGrid.TextMatrix(I, 3) = Format(GesProz, "0.00")

Next I

grdGrid.FixedRows = 1

End Sub

Private Sub grid_color(Zeile As Integer, Farbe As Long)

    Dim Zaehler As Integer

    For Zaehler = 1 To 3

        grdGrid.Col = Zaehler

        grdGrid.Row = Zeile

        grdGrid.CellBackColor = Farbe

    Next Zaehler

End Sub

Private Sub txtA_LostFocus()

    If Not IsNumeric(txtA.Text) Then

        txtA.Text = awert

        txtA.SetFocus

    ElseIf Val(txtA.Text) <= 0 Then

        txtA.Text = awert

        txtA.SetFocus

    Else

        Wertetesten

    End If

End Sub

Private Sub txtB_LostFocus()

    If Not IsNumeric(txtB.Text) Then

        txtB.Text = bwert

        txtB.SetFocus

    ElseIf Val(txtB.Text) <= 0 Then

        txtB.Text = bwert

        txtB.SetFocus

    Else

        Wertetesten

    End If

End Sub

Sub Wertetesten()

Dim fehler As Boolean

Dim ctest As Integer

fehler = False

    If Val(txtA.Text) < Val(txtB.Text) Then

        fehler = True

    End If

    If Val(txtA.Text) + Val(txtB.Text) > 99 Then

        fehler = True

    End If

    ctest = 100 - Val(txtA.Text) - Val(txtB.Text)

    If ctest > Val(txtB.Text) Then

        fehler = True

    End If

If Not fehler Then

    txtC.Text = ctest

    awert = Val(txtA.Text)

    bwert = Val(txtB.Text)

    cwert = ctest

    If AnzahlSaetze > 0 Then Sortieren

Else

    txtA.Text = awert

    txtB.Text = bwert

End If

End Sub

